# Time-Invariant Predictors in Models of Change

The models for this example use the same response time data as in Hoffman (2015) chapter 6, but will include new predictors. Specifically, we will be examining baseline age, abstract reasoning, and education level as time-invariant predictors of change in response time (RT) in milliseconds over six practice sessions to a measure of processing speed (as measured by the number match 3 test) in a sample of 101 older adults.

## SAS Syntax for Data Manipulation:

```
* Defining global variable for file location to be replaced in code below;
%LET filesave= C:\Dropbox\17_CLP944\CLDP944_Example07;
* Location for SAS files for these models (uses macro variable filesave);
LIBNAME filesave "&filesave.";
* Defining macro variable for datafile name to be replaced in code below;
%LET datafile=Example07;

* Bringing data into work library and recoding/centering variables;
* Centering time for polynomial models;
* Creating squared term for use in PROC MEANS only;
DATA &datafile.; SET example.&datafile.;
       * Centering time at session 1 for polynomial models;
       time1 = session - 1; time1sq = time1*time1;
       LABEL time1 = "time1: Session (0=1)";
* Creating two slopes for piecewise models;
            IF Session = 1 THEN DO; Slope12 = 0; Slope26 = 0; END;
       ELSE IF Session = 2 THEN DO; Slope12 = 1; Slope26 = 0; END;
       ELSE IF Session > 2 THEN DO; Slope12 = 1; Slope26 = Session-2; END;
       LABEL Slope12 = "1-2 Early Practice Slope"
             Slope26 = "2-6 Later Practice Slope";
* Centering level-2 predictors;
       Age80  = baseage - 80;
       Reas22 = AbsReas - 22;
       LABEL Age80  = "Age Centered (0=80)"
             Reas22 = "Abstract Reasoning Centered (0=22)";
* Make education a grouping variable FOR DEMO PURPOSES;
            IF EducYrs = . THEN EducGrp = .;
       ELSE IF EducYrs LE 12                    THEN EducGrp=1;
       ELSE IF EducYrs GT 12 AND EducYrs LE 16 THEN EducGrp=2;
       ELSE IF EducYrs GT 16                    THEN EducGrp=3;
       LABEL EducGrp= "Education Group (1=HS, 2=BA, 3=GRAD)"; RUN;

* REMOVING CASES WITH MISSING PREDICTORS OR OUTCOME;
DATA trimmed; SET &datafile.;
       WHERE NMISS(Age80, Reas22, EducGrp, session, nm3rt)=0; RUN;
* Changing dataset used in analyses below;
%LET datafile=trimmed;
```

---

```
* Get variance of level-1 time-related predictors for slope reliability;
PROC MEANS VAR DATA=&datafile.; VAR slope12 slope26 time1 time1sq; RUN;
```

```
The MEANS Procedure

Variable        Variance
------------------------
Slope12        0.1391185
Slope26        2.2258953
time1          2.9214876
time1sq       79.2696970
------------------------
```

Formulas for Intercept Reliability (IR) and Slope Reliability (SR):

$$IR = \frac{\tau^2_{U_0}}{\tau^2_{U_0} + \frac{\sigma^2_e}{L1n}} \qquad SR = \frac{\tau^2_{U_1}}{\tau^2_{U_1} + \frac{\sigma^2_e}{L1n * \sigma^2_{L1}}}$$

## Model 1a. Baseline Unconditional Random Piecewise Growth Model in REML

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$

Level 2: Intercept: $\beta_{0i} = \gamma_{00} + U_{0i}$

Slope12: $\beta_{1i} = \gamma_{10} + U_{1i}$

Slope26: $\beta_{2i} = \gamma_{20} + U_{2i}$

```
TITLE1 "1a: Piecewise Unconditional Model - Random Early/Later Practice Slopes";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=REML;
       CLASS ID session;
       MODEL nm3rt = Slope12 Slope26
                    / SOLUTION DDFM=Satterthwaite OUTPM=PredPUnc; *Save time-predicted RT;;
       RANDOM INTERCEPT Slope12 Slope26 / GCORR TYPE=UN SUBJECT=ID;
       REPEATED session / TYPE=VC SUBJECT=ID;
       ODS OUTPUT CovParms=CovPUnc; * Save covparms for comparison;
RUN; TITLE1;
PROC CORR DATA=PredPUnc OUTP=CorrPUnc; VAR pred; WITH nm3rt; RUN; * Corr of pred and actual RT;
```

Iteration History

| Iteration | Evaluations | -2 Res Log Like | Criterion |
|---|---|---|---|
| 0 | 1 | 9188.48345679 | |
| 1 | 1 | **8275.37431715** | 0.00000000 |

Still use this −2LL for your online homework (it provides 2+ digits after the decimal)

Estimated G Correlation Matrix

| Row | Effect | Participant ID | Col1 | Col2 | Col3 |
|---|---|---|---|---|---|
| 1 | Intercept | 101 | 1.0000 | -0.4025 | -0.3902 |
| 2 | Slope12 | 101 | **-0.4025** | 1.0000 | -0.1293 |
| 3 | Slope26 | 101 | **-0.3902** | **-0.1293** | 1.0000 |

These are the correlations among the random effects.

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr Z | |
|---|---|---|---|---|---|---|
| UN(1,1) | ID | 284312 | 42731 | 6.65 | <.0001 | Random Intercept variance |
| UN(2,1) | ID | -54270 | 18230 | -2.98 | 0.0029 | Int-Slope12 covariance |
| UN(2,2) | ID | 63954 | 13244 | 4.83 | <.0001 | Random Slope12 variance |
| UN(3,1) | ID | -10644 | 3791.26 | -2.81 | 0.0050 | Int-Slope26 covariance |
| UN(3,2) | ID | -1672.30 | 2097.03 | -0.80 | 0.4252 | Slope12-Slope26 covariance |
| UN(3,3) | ID | 2617.28 | 636.48 | 4.11 | <.0001 | Random Slope26 variance |
| session | ID | 17673 | 1435.84 | 12.31 | <.0001 | Residual (e) variance |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| **8275.4** | **7** | 8289.4 | 8289.6 | 8296.8 | 8307.7 | 8314.7 |

Solution for Fixed Effects

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| | |
|---|---|---|---|---|---|---|
| Intercept | 1961.89 | 54.6805 | 100 | 35.88 | <.0001 | Predicted RT when time=0 (session 1 here) |
| Slope12 | -163.64 | 30.2188 | 100 | -5.42 | <.0001 | RT Change/session between sessions 1 and 2 |
| Slope26 | -32.8932 | 6.5888 | 100 | -4.99 | <.0001 | RT Change/session between sessions 2 and 6 |

Pearson Correlation Coefficients, N = 606
Prob > |r| under H0: Rho=0

| | Pred |
|---|---|
| nm3rt | 0.19338 |
| Number Match 3 RT | <.0001 |

r = .19338 → **TOTAL $R^2$ = .0374**
~ 4% of RT variance is accounted for by 2 piecewise linear effects of session

| | ENTER | | | | CALC |
|---|---|---|---|---|---|
| Piecewise Model Term | Random Effect Variance | Residual Variance | L1 Sample Size Per L2 | L1 Predictor Variance | Random Effect Reliability |
| Unc Intercept | 284312 | 17673 | 6 | 1 | 0.990 |
| Unc Slope12 | 63954 | 17673 | 6 | 0.1391185 | 0.751 |
| Unc Slope26 | 2617.28 | 17673 | 6 | 2.2258953 | 0.664 |

Reliability of the unconditional random effects variances (from excel)

---

## Model 1b. Piecewise Model with Fixed Effects of Age on Intercept, Slope12, and Slope26

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}(Slope12_{ti}) + \beta_{2i}(Slope26_{ti}) + e_{ti}$

Level 2: Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01}(Age_i - 80) + U_{0i}$

Slope12: $\beta_{1i} = \gamma_{10} + \gamma_{11}(Age_i - 80) + U_{1i}$

Slope26: $\beta_{2i} = \gamma_{20} + \gamma_{21}(Age_i - 80) + U_{2i}$

```
TITLE1 "1b: Add Fixed Effects for Age on Intercept, Slope12, and Slope26";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=REML;
       CLASS ID session;
       MODEL nm3rt = Slope12 Slope26 Age80 Age80*Slope12 Age80*Slope26
             / SOLUTION DDFM=Satterthwaite OUTPM=PredPAge; * Save fixed-predicted RT;
       RANDOM INTERCEPT Slope12 Slope26 / GCORR TYPE=UN SUBJECT=ID;
       REPEATED session / TYPE=VC SUBJECT=ID;
       ODS OUTPUT CovParms=CovPAge; * Save covparms for comparison;
       CONTRAST "DF=3 Wald Test for Effects of Age"
                Age80 1, Age80*Slope12 1, Age80*Slope26 1 / CHISQ;
       * Requesting slopes at age 80 and 90;
       ESTIMATE "Slope 1-2 for Age 80" Slope12 1 Age80*Slope12 0;
       ESTIMATE "Slope 1-2 for Age 90" Slope12 1 Age80*Slope12 10;
       ESTIMATE "Slope 2-6 for Age 80" Slope26 1 Age80*Slope26 0;
       ESTIMATE "Slope 2-6 for Age 90" Slope26 1 Age80*Slope26 10;
       * Requesting additional effects for age;
       ESTIMATE "Age Effect at Session 1" Age80 1 Age80*Slope12 0 Age80*Slope26 0;
       ESTIMATE "Age Effect at Session 2" Age80 1 Age80*Slope12 1 Age80*Slope26 0;
       ESTIMATE "Age Effect at Session 3" Age80 1 Age80*Slope12 1 Age80*Slope26 1;
       ESTIMATE "Age Effect at Session 4" Age80 1 Age80*Slope12 1 Age80*Slope26 2;
       ESTIMATE "Age Effect at Session 5" Age80 1 Age80*Slope12 1 Age80*Slope26 3;
       ESTIMATE "Age Effect at Session 6" Age80 1 Age80*Slope12 1 Age80*Slope26 4;
RUN; TITLE1;
```

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr Z |
|---|---|---|---|---|---|
| UN(1,1) | ID | 254286 | 38681 | 6.57 | <.0001 |
| UN(2,1) | ID | -46576 | 17140 | -2.72 | 0.0066 |
| UN(2,2) | ID | 62742 | 13139 | 4.78 | <.0001 |
| UN(3,1) | ID | -9251.94 | 3583.89 | -2.58 | 0.0098 |
| UN(3,2) | ID | -2106.57 | 2100.78 | -1.00 | 0.3160 |
| UN(3,3) | ID | 2593.60 | 636.25 | 4.08 | <.0001 |
| session | ID | 17673 | 1435.84 | 12.31 | <.0001 |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| 8251.0 | 7 | 8265.0 | 8265.2 | 8272.4 | 8283.3 | 8290.3 |

Solution for Fixed Effects

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Intercept | 1966.86 | 51.9106 | 99 | 37.89 | <.0001 |
| Slope12 | -164.91 | 30.0311 | 99 | -5.49 | <.0001 |
| Slope26 | -33.1182 | 6.5734 | 99 | -5.04 | <.0001 |
| Age80 | 29.7804 | 8.5822 | 99 | 3.47 | 0.0008 |
| Slope12*Age80 | -7.5810 | 4.9650 | 99 | -1.53 | 0.1300 |
| Slope26*Age80 | -1.3499 | 1.0868 | 99 | -1.24 | 0.2171 |

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Slope 1-2 for Age 80 | -164.91 | 30.0311 | 99 | -5.49 | <.0001 |
| Slope 1-2 for Age 90 | -240.72 | 58.7292 | 99 | -4.10 | <.0001 |
| Slope 2-6 for Age 80 | -33.1182 | 6.5734 | 99 | -5.04 | <.0001 |
| Slope 2-6 for Age 90 | -46.6173 | 12.8551 | 99 | -3.63 | 0.0005 |
| Age Effect at Session 1 | 29.7804 | 8.5822 | 99 | 3.47 | 0.0008 |
| Age Effect at Session 2 | 22.1993 | 7.9689 | 99 | 2.79 | 0.0064 |
| Age Effect at Session 3 | 20.8494 | 7.5245 | 99 | 2.77 | 0.0067 |
| Age Effect at Session 4 | 19.4995 | 7.2176 | 99 | 2.70 | 0.0081 |
| Age Effect at Session 5 | 18.1496 | 7.0663 | 99 | 2.57 | 0.0117 |
| Age Effect at Session 6 | 16.7997 | 7.0805 | 99 | 2.37 | 0.0196 |

Interpret the fixed intercept:

Interpret the fixed effect of Slope12:

Interpret the fixed effect of Slope26:

Interpret the effect of Age80:

Interpret the effect of Slope12*Age80:

Interpret the effect of Slope26*Age80:

Is the age by piecewise model (1b) better than the unconditional piecewise growth model (1a)? How do we know?

Contrasts

| Label | Num DF | Den DF | Chi-Square | F Value | Pr > ChiSq | Pr > F |
|---|---|---|---|---|---|---|
| DF=3 Wald Test for Effects of Age | 3 | 99 | 12.23 | 4.08 | 0.0066 | 0.0089 |

## Syntax and output for Total-$R^2$ macro to compare total $R^2$ values across models:

```
* Calculate Total R2 change relative to unconditional model;
%TotalR2(DV=nm3rt, PredFewer=PredPUnc, PredMore=PredPAge);
```

**Total R2 (% Reduction) for PredPUnc vs. PredPAge**

| Name | Pred Corr | TotalR2 | Total R2Diff |
|---|---|---|---|
| PredPUnc | 0.19338 | 0.03740 | . |
| PredPAge | 0.32795 | 0.10755 | 0.070156 |

## Syntax and output for Pseudo-$R^2$ macro to compare variance components across models:

```
* Calculate PseudoR2 relative to unconditional model;
%PseudoR2(NCov=7, CovFewer=CovPUnc, CovMore=CovPAge);
```

**PsuedoR2 (% Reduction) for CovPUnc vs. CovPAge**

| Name | CovParm | Subject | Estimate | StdErr | ZValue | ProbZ | Pseudo R2 |
|---|---|---|---|---|---|---|---|
| CovPUnc | UN(1,1) | ID | 284312 | 42731 | 6.65 | <.0001 | . |
| CovPUnc | UN(2,2) | ID | 63954 | 13244 | 4.83 | <.0001 | . |
| CovPUnc | UN(3,3) | ID | 2617.28 | 636.48 | 4.11 | <.0001 | . |
| CovPUnc | session | ID | 17673 | 1435.84 | 12.31 | <.0001 | . |
| CovPAge | UN(1,1) | ID | 254286 | 38681 | 6.57 | <.0001 | 0.10561 |
| CovPAge | UN(2,2) | ID | 62742 | 13139 | 4.78 | <.0001 | 0.01895 |
| CovPAge | UN(3,3) | ID | 2593.60 | 636.25 | 4.08 | <.0001 | 0.00905 |
| CovPAge | session | ID | 17673 | 1435.84 | 12.31 | <.0001 | 0.00000 |

Which variance component should have been reduced by each new fixed effect of age?

---

## Model 1c. Piecewise Model with Fixed Effects of Age and Reasoning on Intercept, Slope12, Slope26

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$

Level 2: Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01}(\text{Age}_i - 80) + \gamma_{02}(\text{Reason}_i - 22) + U_{0i}$

Slope12: $\beta_{1i} = \gamma_{10} + \gamma_{11}(\text{Age}_i - 80) + \gamma_{12}(\text{Reason}_i - 22) + U_{1i}$

Slope26: $\beta_{2i} = \gamma_{20} + \gamma_{21}(\text{Age}_i - 80) + \gamma_{22}(\text{Reason}_i - 22) + U_{2i}$

```
TITLE1 "1c: Keep Age, Add Fixed Effects for Reasoning on Intercept, Slope12, and Slope26";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
       CLASS ID session;
       MODEL nm3rt = Slope12 Slope26 Age80 Age80*Slope12 Age80*Slope26
                     Reas22 Reas22*Slope12 Reas22*Slope26
                      / SOLUTION DDFM=Satterthwaite OUTPM=PredPReas; * Save fixed-predicted RT;
       RANDOM INTERCEPT Slope12 Slope26 / GCORR TYPE=UN SUBJECT=ID;
       REPEATED session / TYPE=VC SUBJECT=ID;
       ODS OUTPUT CovParms=CovPReas; * Save covparms for comparison;
       CONTRAST "DF=3 Wald Test for Effects of Age"
                 Age80 1, Age80*Slope12 1, Age80*Slope26 1 / CHISQ;
       CONTRAST "DF=3 Wald Test for Effects of Reasoning"
                Reas22 1, Reas22*Slope12 1, Reas22*Slope26 1 / CHISQ;
       * Requesting additional effects for reasoning;
       ESTIMATE "Reasoning Effect at Session 1" Reas22 1 Reas22*Slope12 0 Reas22*Slope26 0;
       ESTIMATE "Reasoning Effect at Session 2" Reas22 1 Reas22*Slope12 1 Reas22*Slope26 0;
       ESTIMATE "Reasoning Effect at Session 3" Reas22 1 Reas22*Slope12 1 Reas22*Slope26 1;
       ESTIMATE "Reasoning Effect at Session 4" Reas22 1 Reas22*Slope12 1 Reas22*Slope26 2;
       ESTIMATE "Reasoning Effect at Session 5" Reas22 1 Reas22*Slope12 1 Reas22*Slope26 3;
       ESTIMATE "Reasoning Effect at Session 6" Reas22 1 Reas22*Slope12 1 Reas22*Slope26 4;
RUN; TITLE1;
```

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr Z |
|---|---|---|---|---|---|
| UN(1,1) | ID | 242192 | 37151 | 6.52 | <.0001 |
| UN(2,1) | ID | -49817 | 17064 | -2.92 | 0.0035 |
| UN(2,2) | ID | 63222 | 13272 | 4.76 | <.0001 |
| UN(3,1) | ID | -7510.98 | 3414.18 | -2.20 | 0.0278 |
| UN(3,2) | ID | -1845.11 | 2068.67 | -0.89 | 0.3724 |
| UN(3,3) | ID | 2411.55 | 614.00 | 3.93 | <.0001 |
| session | ID | 17673 | 1435.84 | 12.31 | <.0001 |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| 8226.5 | 7 | 8240.5 | 8240.7 | 8247.9 | 8258.8 | 8265.8 |

Solution for Fixed Effects

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Intercept | 1982.64 | 51.1793 | 98 | 38.74 | <.0001 |
| Slope12 | -162.16 | 30.3688 | 98 | -5.34 | <.0001 |
| Slope26 | -35.0669 | 6.4901 | 98 | -5.40 | <.0001 |
| Age80 | 23.0041 | 8.8639 | 98 | 2.60 | 0.0109 |
| Slope12*Age80 | -8.7589 | 5.2597 | 98 | -1.67 | 0.0990 |
| Slope26*Age80 | -0.5135 | 1.1240 | 98 | -0.46 | 0.6488 |
| Reas22 | -27.1200 | 11.4528 | 98 | -2.37 | 0.0198 |
| Slope12*Reas22 | -4.7141 | 6.7959 | 98 | -0.69 | 0.4895 |
| Slope26*Reas22 | 3.3476 | 1.4523 | 98 | 2.30 | 0.0233 |

Which fixed effects are conditional on age?

Which fixed effects are conditional on reasoning?

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Reasoning Effect at Session 1 | -27.1200 | 11.4528 | 98 | -2.37 | 0.0198 |
| Reasoning Effect at Session 2 | -31.8340 | 10.4508 | 98 | -3.05 | 0.0030 |
| Reasoning Effect at Session 3 | -28.4864 | 9.9154 | 98 | -2.87 | 0.0050 |
| Reasoning Effect at Session 4 | -25.1388 | 9.5724 | 98 | -2.63 | 0.0100 |
| Reasoning Effect at Session 5 | -21.7912 | 9.4427 | 98 | -2.31 | 0.0231 |
| Reasoning Effect at Session 6 | -18.4436 | 9.5350 | 98 | -1.93 | 0.0560 |

Contrasts

| Label | Num DF | Den DF | Chi-Square | F Value | Pr > ChiSq | Pr > F |
|---|---|---|---|---|---|---|
| DF=3 Wald Test for Effects of Age | 3 | 98 | 7.10 | 2.37 | 0.0688 | 0.0756 |
| DF=3 Wald Test for Effects of Reasoning | 3 | 98 | 10.51 | 3.50 | 0.0147 | 0.0183 |



### Syntax and output from additional macros for effect size:

```
* Calculate Total R2 change relative to age only model;
%TotalR2(DV=nm3rt, PredFewer=PredPAge, PredMore=PredPReas);
```

```
Total R2 (% Reduction) for PredPAge vs. PredPReas
              Pred                 Total
  Name        Corr      TotalR2    R2Diff
PredPAge     0.32795    0.10755     .
PredPReas    0.40163    0.16131    0.053755
```

```
* Calculate PseudoR2 relative to age only model;
%PseudoR2(NCov=7, CovFewer=CovPAge, CovMore=CovPReas);
```

```
PsuedoR2 (% Reduction) for CovPAge vs. CovPReas
  Name      CovParm    Subject    Estimate      StdErr    ZValue     ProbZ     PseudoR2
CovPAge     UN(1,1)      ID         254286       38681      6.57    <.0001      .
CovPAge     UN(2,2)      ID          62742       13139      4.78    <.0001      .
CovPAge     UN(3,3)      ID        2593.60      636.25      4.08    <.0001      .
CovPAge     session      ID          17673     1435.84     12.31    <.0001      .
CovPReas    UN(1,1)      ID         242192       37151      6.52    <.0001     0.047560
CovPReas    UN(2,2)      ID          63222       13272      4.76    <.0001    -0.007643
CovPReas    UN(3,3)      ID        2411.55      614.00      3.93    <.0001     0.070193
CovPReas    session      ID          17673     1435.84     12.31    <.0001    -0.000000
```

Which variance component should have been reduced by each new fixed effect of age?

---

### Model 1d. Piecewise Model Adding Education Group on Intercept, Slope12, Slope26

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$

Level 2:

Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01}(\text{Age}_i - 80) + \gamma_{02}(\text{Reason}_i - 22) + \gamma_{03}(\text{Highvs.LowEd}_i) + \gamma_{04}(\text{Highvs.MedEd}_i) + U_{0i}$

Slope12: $\beta_{1i} = \gamma_{10} + \gamma_{11}(\text{Age}_i - 80) + \gamma_{12}(\text{Reason}_i - 22) + \gamma_{13}(\text{Highvs.LowEd}_i) + \gamma_{14}(\text{Highvs.MedEd}_i) + U_{1i}$

Slope26: $\beta_{2i} = \gamma_{20} + \gamma_{21}(\text{Age}_i - 80) + \gamma_{22}(\text{Reason}_i - 22) + \gamma_{23}(\text{Highvs.LowEd}_i) + \gamma_{24}(\text{Highvs.MedEd}_i) + U_{2i}$

```
TITLE1 "1d: Keep Age & Reasoning, Add Education Group on Intercept, Slope12, and Slope26";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=REML;
       CLASS ID EducGrp session;
       MODEL nm3rt = Slope12 Slope26 Age80 Age80*Slope12 Age80*Slope26
                     Reas22 Reas22*Slope12 Reas22*Slope26
                     EducGrp Slope12*EducGrp Slope26*EducGrp
                     / SOLUTION DDFM=Satterthwaite OUTPM=PredPEduc; * Save fixed-predicted RT;
       RANDOM INTERCEPT Slope12 Slope26 / GCORR TYPE=UN SUBJECT=ID;
       REPEATED session / TYPE=VC SUBJECT=ID;
       ODS OUTPUT CovParms=CovPEduc; * Save covparms for comparison;
       CONTRAST "DF=3 Wald Test for Effects of Age"
                Age80 1, Age80*Slope12 1, Age80*Slope26 1 / CHISQ;
       CONTRAST "DF=3 Wald Test for Effects of Reasoning"
                Reas22 1, Reas22*Slope12 1, Reas22*Slope26 1 / CHISQ;
       CONTRAST "DF=6 Wald Test for Effects of Education"
                EducGrp -1 1 0, EducGrp -1 0 1, EducGrp*Slope12 -1 1 0, EducGrp*Slope12 -1 0 1,
                EducGrp*Slope26 -1 1 0, EducGrp*Slope26 -1 0 1 / CHISQ;
       * LSMEANS gives follow-up tests and means per group for education main effect only;
       LSMEANS EducGrp / AT (Slope12 Slope26 Age80 Reas22) = (0 0 0 0) DIFF=ALL;
       LSMEANS EducGrp / AT (Slope12 Slope26 Age80 Reas22) = (1 4 0 0) DIFF=ALL;
```

```
      * ESTIMATE statements can also give specific effects as before;
      ESTIMATE "L vs. H Educ for Intercept "  EducGrp -1  0  1 ;
      ESTIMATE "M vs. H Educ for Intercept"   EducGrp  0 -1  1 ;
      ESTIMATE "L vs. M Educ for Intercept"   EducGrp -1  1  0 ;
      ESTIMATE "L vs. H Educ for Slope12"     Slope12*EducGrp -1  0  1 ;
      ESTIMATE "M vs. H Educ for Slope12"     Slope12*EducGrp  0 -1  1 ;
      ESTIMATE "L vs. M Educ for Slope12"     Slope12*EducGrp -1  1  0 ;
      ESTIMATE "L vs. H Educ for Slope26"     Slope26*EducGrp -1  0  1 ;
      ESTIMATE "M vs. H Educ for Slope26"     Slope26*EducGrp  0 -1  1 ;
      ESTIMATE "L vs. M Educ for Slope26"     Slope26*EducGrp -1  1  0 ;
RUN; TITLE1;
```

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr Z |
|---|---|---|---|---|---|
| UN(1,1) | ID | 246920 | 38218 | 6.46 | <.0001 |
| UN(2,1) | ID | -52254 | 17491 | -2.99 | 0.0028 |
| UN(2,2) | ID | 63495 | 13444 | 4.72 | <.0001 |
| UN(3,1) | ID | -7543.48 | 3493.64 | -2.16 | 0.0308 |
| UN(3,2) | ID | -1820.21 | 2099.75 | -0.87 | 0.3860 |
| UN(3,3) | ID | 2446.05 | 624.87 | 3.91 | <.0001 |
| session | ID | 17673 | 1435.84 | 12.31 | <.0001 |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| 8164.2 | 7 | 8178.2 | 8178.4 | 8185.6 | 8196.5 | 8203.5 |

Solution for Fixed Effects

| Effect | Education Group (1=HS,2=BA,3=GRAD) | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|---|
| Intercept | | 1978.15 | 105.83 | 96 | 18.69 | <.0001 |
| Slope12 | | -153.14 | 62.3250 | 96 | -2.46 | 0.0158 |
| Slope26 | | -24.6403 | 13.3543 | 96 | -1.85 | 0.0681 |
| Age80 | | 22.9367 | 8.9490 | 96 | 2.56 | 0.0119 |
| Slope12*Age80 | | -8.9054 | 5.2704 | 96 | -1.69 | 0.0943 |
| Slope26*Age80 | | -0.5289 | 1.1293 | 96 | -0.47 | 0.6406 |
| Reas22 | | -28.5673 | 11.9710 | 96 | -2.39 | 0.0190 |
| Slope12*Reas22 | | -7.0891 | 7.0501 | 96 | -1.01 | 0.3172 |
| Slope26*Reas22 | | 3.4883 | 1.5106 | 96 | 2.31 | 0.0231 |
| EducGrp | 1 | -41.9718 | 157.35 | 96 | -0.27 | 0.7902 |
| EducGrp | 2 | 25.4470 | 125.54 | 96 | 0.20 | 0.8398 |
| EducGrp | 3 | 0 | . | . | . | . |
| Slope12*EducGrp | 1 | -85.9455 | 92.6714 | 96 | -0.93 | 0.3560 |
| Slope12*EducGrp | 2 | 18.5834 | 73.9371 | 96 | 0.25 | 0.8021 |
| Slope12*EducGrp | 3 | 0 | . | . | . | . |
| Slope26*EducGrp | 1 | -6.3237 | 19.8566 | 96 | -0.32 | 0.7508 |
| Slope26*EducGrp | 2 | -16.5965 | 15.8424 | 96 | -1.05 | 0.2975 |
| Slope26*EducGrp | 3 | 0 | . | . | . | . |

Type 3 Tests of Fixed Effects

| Effect | Num DF | Den DF | F Value | Pr > F |
|---|---|---|---|---|
| Slope12 | 1 | 96 | 28.16 | <.0001 |
| Slope26 | 1 | 96 | 20.73 | <.0001 |
| Age80 | 1 | 96 | 6.57 | 0.0119 |
| Slope12*Age80 | 1 | 96 | 2.86 | 0.0943 |
| Slope26*Age80 | 1 | 96 | 0.22 | 0.6406 |
| Reas22 | 1 | 96 | 5.69 | 0.0190 |
| Slope12*Reas22 | 1 | 96 | 1.01 | 0.3172 |
| Slope26*Reas22 | 1 | 96 | 5.33 | 0.0231 |
| **EducGrp** | **2** | **96** | **0.12** | **0.8831** |
| **Slope12*EducGrp** | **2** | **96** | **0.85** | **0.4289** |
| **Slope26*EducGrp** | **2** | **96** | **0.60** | **0.5516** |

I normally skip this box if the CLASS statement is not used for predictors, but here the last three entries give us the omnibus (df=2) tests for whether there are any education group differences on the intercept, slope12, or slope26 time slopes, not just pairwise comparisons.

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| L vs. H Educ for Intercept | 41.9718 | 157.35 | 96 | 0.27 | 0.7902 |
| M vs. H Educ for Intercept | -25.4470 | 125.54 | 96 | -0.20 | 0.8398 |
| L vs. M Educ for Intercept | 67.4187 | 136.36 | 96 | 0.49 | 0.6221 |
| L vs. H Educ for Slope12 | 85.9455 | 92.6714 | 96 | 0.93 | 0.3560 |
| M vs. H Educ for Slope12 | -18.5834 | 73.9371 | 96 | -0.25 | 0.8021 |
| L vs. M Educ for Slope12 | 104.53 | 80.3066 | 96 | 1.30 | 0.1962 |
| L vs. H Educ for Slope26 | 6.3237 | 19.8566 | 96 | 0.32 | 0.7508 |
| M vs. H Educ for Slope26 | 16.5965 | 15.8424 | 96 | 1.05 | 0.2975 |
| L vs. M Educ for Slope26 | -10.2728 | 17.2072 | 96 | -0.60 | 0.5519 |

Contrasts

| Label | Num DF | Den DF | Chi-Square | F Value | Pr > ChiSq | Pr > F |
|---|---|---|---|---|---|---|
| DF=3 Wald Test for Effects of Age | 3 | 96 | 6.96 | 2.32 | 0.0732 | 0.0802 |
| DF=3 Wald Test for Effects of Reasoning | 3 | 96 | 11.92 | 3.97 | 0.0077 | 0.0103 |
| DF=6 Wald Test for Effects of Education | 6 | 96 | 4.38 | 0.73 | 0.6252 | 0.6264 |

Least Squares Means

| Effect | Education Group (1=HS, 2=BA,3=GRAD) | Slope12 | Slope26 | Age80 | Reas22 | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|---|---|---|---|---|
| EducGrp | 1 | 0.00 | 0.00 | 0.00 | 0.00 | 1936.18 | 114.13 | 96 | 16.97 | <.0001 |
| EducGrp | 2 | 0.00 | 0.00 | 0.00 | 0.00 | 2003.60 | 70.3593 | 96 | 28.48 | <.0001 |
| EducGrp | 3 | 0.00 | 0.00 | 0.00 | 0.00 | 1978.15 | 105.83 | 96 | 18.69 | <.0001 |
| EducGrp | 1 | 1.00 | 4.00 | 0.00 | 0.00 | 1573.24 | 94.3228 | 96 | 16.68 | <.0001 |
| EducGrp | 2 | 1.00 | 4.00 | 0.00 | 0.00 | 1704.10 | 58.1509 | 96 | 29.30 | <.0001 |
| EducGrp | 3 | 1.00 | 4.00 | 0.00 | 0.00 | 1726.45 | 87.4643 | 96 | 19.74 | <.0001 |

Differences of Least Squares Means

| Effect | (1=HS, 2=BA, 3=GRAD) | (1=HS, 2=BA, 3=GRAD) | Slope12 | Slope26 | Age80 | Reas22 | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|---|---|---|---|---|---|
| EducGrp | 1 | 2 | 0.00 | 0.00 | 0.00 | 0.00 | -67.4187 | 136.36 | 96 | -0.49 | 0.6221 |
| EducGrp | 1 | 3 | 0.00 | 0.00 | 0.00 | 0.00 | -41.9718 | 157.35 | 96 | -0.27 | 0.7902 |
| EducGrp | 2 | 3 | 0.00 | 0.00 | 0.00 | 0.00 | 25.4470 | 125.54 | 96 | 0.20 | 0.8398 |
| EducGrp | 1 | 2 | 1.00 | 4.00 | 0.00 | 0.00 | -130.86 | 112.70 | 96 | -1.16 | 0.2485 |
| EducGrp | 1 | 3 | 1.00 | 4.00 | 0.00 | 0.00 | -153.21 | 130.05 | 96 | -1.18 | 0.2417 |
| EducGrp | 2 | 3 | 1.00 | 4.00 | 0.00 | 0.00 | -22.3558 | 103.76 | 96 | -0.22 | 0.8299 |

In LSMEANS, you must specify a value at which to hold each continuous predictor.

## Syntax and output from additional macros for effect size:

```
* Calculate Total R2 change relative to model with age and reasoning;
%TotalR2(DV=nm3rt, PredFewer=PredPReas, PredMore=PredPEduc);
```

**Total R2 (% Reduction) for PredPReas vs. PredPEduc**

| Name | Pred Corr | TotalR2 | Total R2Diff |
|---|---|---|---|
| PredPReas | 0.40163 | 0.16131 | . |
| PredPEduc | 0.41669 | 0.17363 | 0.012322 |

```
* Calculate PseudoR2 relative to model with age and reasoning;
%PseudoR2(NCov=7, CovFewer=CovPReas, CovMore=CovPEduc);
```

**PsuedoR2 (% Reduction) for CovPReas vs. CovPEduc**

| Name | CovParm | Subject | Estimate | StdErr | ZValue | ProbZ | PseudoR2 |
|---|---|---|---|---|---|---|---|
| CovPReas | UN(1,1) | ID | 242192 | 37151 | 6.52 | <.0001 | . |
| CovPReas | UN(2,2) | ID | 63222 | 13272 | 4.76 | <.0001 | . |
| CovPReas | UN(3,3) | ID | 2411.55 | 614.00 | 3.93 | <.0001 | . |
| CovPReas | session | ID | 17673 | 1435.84 | 12.31 | <.0001 | . |
| CovPEduc | UN(1,1) | ID | 246920 | 38218 | 6.46 | <.0001 | -0.019521 |
| CovPEduc | UN(2,2) | ID | 63495 | 13444 | 4.72 | <.0001 | -0.004322 |
| CovPEduc | UN(3,3) | ID | 2446.05 | 624.87 | 3.91 | <.0001 | -0.014309 |
| CovPEduc | session | ID | 17673 | 1435.84 | 12.31 | <.0001 | 0.000000 |

Given that education group has no significant effects, we can drop it entirely before moving on to examine potential interactions among the time-invariant predictors of baseline age and reasoning.

---

### Model 1e. Piecewise Model with Age*Reasoning on Intercept, Slope12, Slope26

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Slope12}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$

Level 2:

Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01}(\text{Age}_i - 80) + \gamma_{02}(\text{Reason}_i - 22) + \gamma_{03}(\text{Age}_i - 80)(\text{Reason}_i - 22) + U_{0i}$

Slope12: $\beta_{1i} = \gamma_{10} + \gamma_{11}(\text{Age}_i - 80) + \gamma_{12}(\text{Reason}_i - 22) + \gamma_{13}(\text{Age}_i - 80)(\text{Reason}_i - 22) + U_{1i}$

Slope26: $\beta_{2i} = \gamma_{20} + \gamma_{21}(\text{Age}_i - 80) + \gamma_{22}(\text{Reason}_i - 22) + \gamma_{23}(\text{Age}_i - 80)(\text{Reason}_i - 22) + U_{2i}$

```
TITLE1 "1e: Drop EducGrp, Add Age*Reasoning on Intercept, Slope12, and Slope26";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
       CLASS ID session;
       MODEL nm3rt = Slope12 Slope26 Age80 Age80*Slope12 Age80*Slope26
                     Reas22 Reas22*Slope12 Reas22*Slope26
                     Age80*Reas22 Age80*Reas22*Slope12 Age80*Reas22*Slope26
                     / SOLUTION DDFM=Satterthwaite OUTPM=PredPAgeReas; * Save fixed-predicted RT;
       RANDOM INTERCEPT Slope12 Slope26 / GCORR TYPE=UN SUBJECT=ID;
       REPEATED session / TYPE=VC SUBJECT=ID;
       ODS OUTPUT CovParms=CovPAgeReas; * Save covparms for comparison;
       CONTRAST "DF=3 Wald Test for Effects of Age*Reasoning"
                Age80*Reas22 1, Age80*Reas22*Slope12 1, Age80*Reas22*Slope26 1 / CHISQ;
* Age simple effects to decompose interactions;
ESTIMATE "Age Effect on Intercept, Reas 17"  Age80 1 Reas22*Age80 -5;
ESTIMATE "Age Effect on Intercept, Reas 22"  Age80 1 Reas22*Age80  0;
ESTIMATE "Age Effect on Intercept, Reas 27"  Age80 1 Reas22*Age80  5;
ESTIMATE "Age Effect on Slope12,   Reas 17"  Age80*Slope12 1 Reas22*Age80*Slope12 -5;
ESTIMATE "Age Effect on Slope12,   Reas 22"  Age80*Slope12 1 Reas22*Age80*Slope12  0;
ESTIMATE "Age Effect on Slope12,   Reas 27"  Age80*Slope12 1 Reas22*Age80*Slope12  5;
ESTIMATE "Age Effect on Slope26,   Reas 17"  Age80*Slope26 1 Reas22*Age80*Slope26 -5;
ESTIMATE "Age Effect on Slope26,   Reas 22"  Age80*Slope26 1 Reas22*Age80*Slope26  0;
ESTIMATE "Age Effect on Slope26,   Reas 27"  Age80*Slope26 1 Reas22*Age80*Slope26  5;
* Reasoning simple effects to decompose interactions;
ESTIMATE "Reasoning Effect on Intercept, Age 70"  Reas22 1 Reas22*Age80 -10;
ESTIMATE "Reasoning Effect on Intercept, Age 80"  Reas22 1 Reas22*Age80   0;
ESTIMATE "Reasoning Effect on Intercept, Age 90"  Reas22 1 Reas22*Age80  10;
ESTIMATE "Reasoning Effect on Slope12,   Age 70"  Reas22*Slope12 1 Reas22*Age80*Slope12 -10;
ESTIMATE "Reasoning Effect on Slope12,   Age 80"  Reas22*Slope12 1 Reas22*Age80*Slope12   0;
ESTIMATE "Reasoning Effect on Slope12,   Age 90"  Reas22*Slope12 1 Reas22*Age80*Slope12  10;
ESTIMATE "Reasoning Effect on Slope26,   Age 70"  Reas22*Slope26 1 Reas22*Age80*Slope26 -10;
ESTIMATE "Reasoning Effect on Slope26,   Age 80"  Reas22*Slope26 1 Reas22*Age80*Slope26   0;
ESTIMATE "Reasoning Effect on Slope26,   Age 90"  Reas22*Slope26 1 Reas22*Age80*Slope26  10;
RUN; TITLE1;
```

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr Z |
|---|---|---|---|---|---|
| UN(1,1) | ID | 244192 | 37629 | 6.49 | <.0001 |
| UN(2,1) | ID | -49617 | 17179 | -2.89 | 0.0039 |
| UN(2,2) | ID | 62984 | 13304 | 4.73 | <.0001 |
| UN(3,1) | ID | -7513.67 | 3457.96 | -2.17 | 0.0298 |
| UN(3,2) | ID | -1999.16 | 2088.67 | -0.96 | 0.3385 |
| UN(3,3) | ID | 2446.40 | 621.86 | 3.93 | <.0001 |
| session | ID | 17673 | 1435.84 | 12.31 | <.0001 |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| 8220.9 | 7 | 8234.9 | 8235.1 | 8242.3 | 8253.2 | 8260.2 |

Solution for Fixed Effects

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Intercept | 1974.57 | 53.8381 | 97 | 36.68 | <.0001 |
| Slope12 | -151.52 | 31.7828 | 97 | -4.77 | <.0001 |
| Slope26 | -34.1783 | 6.8294 | 97 | -5.00 | <.0001 |
| Age80 | 22.7598 | 8.9112 | 97 | 2.55 | 0.0122 |
| Slope12*Age80 | -8.4366 | 5.2607 | 97 | -1.60 | 0.1120 |
| Slope26*Age80 | -0.4866 | 1.1304 | 97 | -0.43 | 0.6678 |
| Reas22 | -28.0448 | 11.6437 | 97 | -2.41 | 0.0179 |
| Slope12*Reas22 | -3.4941 | 6.8738 | 97 | -0.51 | 0.6124 |
| Slope26*Reas22 | 3.4494 | 1.4770 | 97 | 2.34 | 0.0216 |
| Age80*Reas22 | -0.9317 | 1.8579 | 97 | -0.50 | 0.6172 |
| Slope12*Age80*Reas22 | 1.2290 | 1.0968 | 97 | 1.12 | 0.2652 |
| Slope26*Age80*Reas22 | 0.1026 | 0.2357 | 97 | 0.44 | 0.6644 |

Which fixed effects are now conditional on age?

Which fixed effects are now conditional on reasoning?

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Age Effect on Intercept, Reas 17 | 27.4184 | 12.5162 | 97 | 2.19 | 0.0309 |
| Age Effect on Intercept, Reas 22 | 22.7598 | 8.9112 | 97 | 2.55 | 0.0122 |
| Age Effect on Intercept, Reas 27 | 18.1011 | 13.2197 | 97 | 1.37 | 0.1741 |
| Age Effect on Slope12, Reas 17 | -14.5818 | 7.3888 | 97 | -1.97 | 0.0513 |
| Age Effect on Slope12, Reas 22 | -8.4366 | 5.2607 | 97 | -1.60 | 0.1120 |
| Age Effect on Slope12, Reas 27 | -2.2914 | 7.8042 | 97 | -0.29 | 0.7697 |
| Age Effect on Slope26, Reas 17 | -0.9994 | 1.5877 | 97 | -0.63 | 0.5305 |
| Age Effect on Slope26, Reas 22 | -0.4866 | 1.1304 | 97 | -0.43 | 0.6678 |
| Age Effect on Slope26, Reas 27 | 0.02627 | 1.6769 | 97 | 0.02 | 0.9875 |
| Reasoning Effect on Intercept, Age 70 | -18.7275 | 20.3038 | 97 | -0.92 | 0.3586 |
| Reasoning Effect on Intercept, Age 80 | -28.0448 | 11.6437 | 97 | -2.41 | 0.0179 |
| Reasoning Effect on Intercept, Age 90 | -37.3622 | 23.4371 | 97 | -1.59 | 0.1142 |
| Reasoning Effect on Slope12, Age 70 | -15.7845 | 11.9862 | 97 | -1.32 | 0.1910 |
| Reasoning Effect on Slope12, Age 80 | -3.4941 | 6.8738 | 97 | -0.51 | 0.6124 |
| Reasoning Effect on Slope12, Age 90 | 8.7963 | 13.8359 | 97 | 0.64 | 0.5264 |
| Reasoning Effect on Slope26, Age 70 | 2.4237 | 2.5756 | 97 | 0.94 | 0.3490 |
| Reasoning Effect on Slope26, Age 80 | 3.4494 | 1.4770 | 97 | 2.34 | 0.0216 |
| Reasoning Effect on Slope26, Age 90 | 4.4751 | 2.9730 | 97 | 1.51 | 0.1355 |

Contrasts

| Label | Num DF | Den DF | Chi-Square | F Value | Pr > ChiSq | Pr > F |
|---|---|---|---|---|---|---|
| DF=3 Wald Test for Age*Reasoning | 3 | 97 | 1.98 | 0.66 | 0.5771 | 0.5791 |

## Syntax and output from additional macros for effect size:

```
* Calculate Total R2 change relative to age and reasoning main effects model;
%TotalR2(DV=nm3rt, PredFewer=PredPReas, PredMore=PredPAgeReas);
```

**Total R2 (% Reduction) for PredPReas vs. PredPAgeReas**

| Name | Pred Corr | TotalR2 | Total R2Diff |
|---|---|---|---|
| PredPReas | 0.40163 | 0.16131 | . |
| PredPAgeReas | 0.40306 | 0.16246 | .001148258 |

```
* Calculate PseudoR2 relative to age and reasoning main effects model;
%PseudoR2(NCov=7, CovFewer=CovPReas, CovMore=CovPAgeReas);
```

**PsuedoR2 (% Reduction) for CovPReas vs. CovPAgeReas**

| Name | CovParm | Subject | Estimate | StdErr | ZValue | ProbZ | PseudoR2 |
|---|---|---|---|---|---|---|---|
| CovPReas | UN(1,1) | ID | 242192 | 37151 | 6.52 | <.0001 | . |
| CovPReas | UN(2,2) | ID | 63222 | 13272 | 4.76 | <.0001 | . |
| CovPReas | UN(3,3) | ID | 2411.55 | 614.00 | 3.93 | <.0001 | . |
| CovPReas | session | ID | 17673 | 1435.84 | 12.31 | <.0001 | . |
| CovPAgeReas | UN(1,1) | ID | 244192 | 37629 | 6.49 | <.0001 | -0.008258 |
| CovPAgeReas | UN(2,2) | ID | 62984 | 13304 | 4.73 | <.0001 | 0.003765 |
| CovPAgeReas | UN(3,3) | ID | 2446.40 | 621.86 | 3.93 | <.0001 | -0.014453 |
| CovPAgeReas | session | ID | 17673 | 1435.84 | 12.31 | <.0001 | 0.000000 |

Based on the lack of significance of the higher-order interactions, I'd say we're done with this model. Age and reasoning as main effects in predicting intercept, slope12, and slope26 seems to be the best model.

---

## Model 2a. Baseline Unconditional Random Quadratic Growth Model in REML

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Session}_{ti} - 1\right) + \beta_{2i}\left(\text{Session}_{ti} - 1\right)^2 + e_{ti}$

Level 2:

Intercept: $\beta_{0i} = \gamma_{00} + U_{0i}$

Linear: $\beta_{1i} = \gamma_{10} + U_{1i}$

Quadratic: $\beta_{2i} = \gamma_{20} + U_{2i}$

```
TITLE1 "2a: Random Quadratic Unconditional Model";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=REML;
       CLASS ID session;
       MODEL nm3rt = time1 time1*time1
                     / SOLUTION DDFM=Satterthwaite OUTPM=PredQUnc; * Save fixed-predicted RT;
       RANDOM INTERCEPT time1 time1*time1 / GCORR TYPE=UN SUBJECT=ID;
       REPEATED session / TYPE=VC SUBJECT=ID;
       ODS OUTPUT CovParms=CovQUnc; * Save covparms for comparison;
RUN; TITLE1;
PROC CORR DATA=PredQUnc OUTP=CorrQUnc; VAR pred; WITH nm3rt; RUN; * Corr of predicted, actual RT;
```

Iteration History

| Iteration | Evaluations | -2 Res Log Like | Criterion |
|---|---|---|---|
| 0 | 1 | 9193.25780414 | |
| 1 | 1 | 8302.74566856 | 0.00000000 |

Still use this −2LL for your online homework (it provides 2+ digits after the decimal)

Estimated G Correlation Matrix

| Row | Effect | Participant ID | Col1 | Col2 | Col3 |
|---|---|---|---|---|---|
| 1 | Intercept | 101 | 1.0000 | -0.4230 | 0.2948 |
| 2 | time1 | 101 | **-0.4230** | 1.0000 | -0.9640 |
| 3 | time1*time1 | 101 | **0.2948** | **-0.9640** | 1.0000 |

These are the correlations among the random effects. Note the strong correlation among linear (at time 0) and quadratic change.

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr Z | |
|---|---|---|---|---|---|---|
| UN(1,1) | ID | 276206 | 41442 | 6.66 | <.0001 | Random Intercept variance |
| UN(2,1) | ID | -35734 | 11941 | -2.99 | 0.0028 | Int-Linear covariance |
| UN(2,2) | ID | 25840 | 5864.41 | 4.41 | <.0001 | Random Linear Slope variance |
| UN(3,1) | ID | 3901.96 | 1949.06 | 2.00 | 0.0453 | Int-Quadratic covariance |
| UN(3,2) | ID | -3903.32 | 982.61 | -3.97 | <.0001 | Linear-Quadratic covariance |
| UN(3,3) | ID | 634.47 | 172.37 | 3.68 | 0.0001 | Random Quadratic Slope variance |
| session | ID | 20298 | 1649.11 | 12.31 | <.0001 | Residual (e) variance |

```
                    Information Criteria
Neg2LogLike     Parms        AIC       AICC       HQIC        BIC       CAIC
     8302.7         7     8316.7     8316.9     8324.2     8335.1     8342.1

                  Solution for Fixed Effects
                            Standard
Effect          Estimate       Error      DF    t Value    Pr > |t|
Intercept       1945.85      53.8497     100      36.13      <.0001  Predicted RT when time=0 (session 1 here)
time1           -120.90      20.0476     100      -6.03      <.0001  RT change/session at session=1
time1*time1     13.8656       3.4154     100       4.06      <.0001  Half rate change in linear slope/session

Pearson Correlation Coefficients, N = 606
        Prob > |r| under H0: Rho=0
                             Pred
nm3rt                     0.19167
Number Match 3 RT          <.0001
```

r = .19167 → **TOTAL $R^2$ = .0367**
~ 4% of RT variance is accounted for by linear and quadratic effects of session

| | ENTER | | | | CALC |
|---|---|---|---|---|---|
| **Quadratic Model Term** | **Random Effect Variance** | **Residual Variance** | **L1 Sample Size Per L2** | **L1 Predictor Variance** | **Random Effect Reliability** |
| Unc Intercept | 276206 | 20298 | 6 | 1 | 0.988 |
| Unc Linear | 25840 | 20298 | 6 | 2.9214876 | 0.957 |
| Unc Quadratic | 634.47 | 20298 | 6 | 79.269697 | 0.937 |

Reliability of the unconditional random effects variances (from excel)

---

## Model 1b. Quadratic Model with Age Predicting Intercept, Linear, Quadratic Time Slopes

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Session}_{ti} - 1) + \beta_{2i}(\text{Session}_{ti} - 1)^2 + e_{ti}$

Level 2:

Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01}(\text{Age}_i - 80) + U_{0i}$

Linear: $\beta_{1i} = \gamma_{10} + \gamma_{11}(\text{Age}_i - 80) + U_{1i}$

Quadratic: $\beta_{2i} = \gamma_{20} + \gamma_{21}(\text{Age}_i - 80) + U_{2i}$

```
TITLE1 "2b: Add Fixed Effects for Age on Intercept, Linear, and Quadratic Time Slopes";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=REML;
       CLASS ID session;
       MODEL nm3rt = time1 time1*time1 Age80 time1*Age80 time1*time1*Age80
                   / SOLUTION DDFM=Satterthwaite OUTPM=PredQAge; * Save fixed-predicted RT;
       RANDOM INTERCEPT time1 time1*time1 / GCORR TYPE=UN SUBJECT=ID;
       REPEATED session / TYPE=VC SUBJECT=ID;
       ODS OUTPUT CovParms=CovQAge; * Save covparms for comparison;
       CONTRAST "DF=3 Wald Test for Effects of Age"
                Age80 1, Age80*time1 1, Age80*time1*time1 1 / CHISQ;
* Requesting additional effects for age;
ESTIMATE "Age Effect at Session 1"    Age80 1 time1*Age80 0 time1*time1*Age80 0;
ESTIMATE "Age Effect at Session 2"    Age80 1 time1*Age80 1 time1*time1*Age80 1;
ESTIMATE "Age Effect at Session 3"    Age80 1 time1*Age80 2 time1*time1*Age80 4;
ESTIMATE "Age Effect at Session 4"    Age80 1 time1*Age80 3 time1*time1*Age80 9;
ESTIMATE "Age Effect at Session 5"    Age80 1 time1*Age80 4 time1*time1*Age80 16;
ESTIMATE "Age Effect at Session 6"    Age80 1 time1*Age80 5 time1*time1*Age80 25;
ESTIMATE "Age*Linear Time Slope at Session 1" time1*Age80 1 time1*time1*Age80 0;
ESTIMATE "Age*Linear Time Slope at Session 2" time1*Age80 1 time1*time1*Age80 2;
ESTIMATE "Age*Linear Time Slope at Session 3" time1*Age80 1 time1*time1*Age80 4;
ESTIMATE "Age*Linear Time Slope at Session 4" time1*Age80 1 time1*time1*Age80 6;
```

```
ESTIMATE "Age*Linear Time Slope at Session 5" time1*Age80 1 time1*time1*Age80 8;
ESTIMATE "Age*Linear Time Slope at Session 6" time1*Age80 1 time1*time1*Age80 10;
RUN; TITLE1;
```

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr Z |
|---|---|---|---|---|---|
| UN(1,1) | ID | 247691 | 37599 | 6.59 | <.0001 |
| UN(2,1) | ID | -30154 | 11191 | -2.69 | 0.0070 |
| UN(2,2) | ID | 25083 | 5787.37 | 4.33 | <.0001 |
| UN(3,1) | ID | 3232.78 | 1847.12 | 1.75 | 0.0801 |
| UN(3,2) | ID | -3830.21 | 976.76 | -3.92 | <.0001 |
| UN(3,3) | ID | 629.58 | 172.51 | 3.65 | 0.0001 |
| session | ID | 20298 | 1649.11 | 12.31 | <.0001 |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| 8283.2 | 7 | 8297.2 | 8297.3 | 8304.6 | 8315.5 | 8322.5 |

Solution for Fixed Effects

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Intercept | 1950.69 | 51.1806 | 99 | 38.11 | <.0001 |
| time1 | -121.83 | 19.8672 | 99 | -6.13 | <.0001 |
| time1*time1 | 13.9774 | 3.4096 | 99 | 4.10 | <.0001 |
| Age80 | 29.0495 | 8.4616 | 99 | 3.43 | 0.0009 |
| time1*Age80 | -5.5946 | 3.2846 | 99 | -1.70 | 0.0916 |
| time1*time1*Age80 | 0.6709 | 0.5637 | 99 | 1.19 | 0.2368 |

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Age Effect at Session 1 | 29.0495 | 8.4616 | 99 | 3.43 | 0.0009 |
| Age Effect at Session 2 | 24.1258 | 7.6862 | 99 | 3.14 | 0.0022 |
| Age Effect at Session 3 | 20.5439 | 7.5343 | 99 | 2.73 | 0.0076 |
| Age Effect at Session 4 | 18.3038 | 7.4038 | 99 | 2.47 | 0.0151 |
| Age Effect at Session 5 | 17.4056 | 7.1425 | 99 | 2.44 | 0.0166 |
| Age Effect at Session 6 | 17.8492 | 7.1254 | 99 | 2.51 | 0.0139 |
| Age*Linear Time Slope at Session 1 | -5.5946 | 3.2846 | 99 | -1.70 | 0.0916 |
| Age*Linear Time Slope at Session 2 | -4.2528 | 2.2283 | 99 | -1.91 | 0.0592 |
| Age*Linear Time Slope at Session 3 | -2.9110 | 1.2977 | 99 | -2.24 | 0.0271 |
| Age*Linear Time Slope at Session 4 | -1.5692 | 0.9720 | 99 | -1.61 | 0.1096 |
| Age*Linear Time Slope at Session 5 | -0.2273 | 1.6576 | 99 | -0.14 | 0.8912 |
| Age*Linear Time Slope at Session 6 | 1.1145 | 2.6632 | 99 | 0.42 | 0.6765 |

Interpret the fixed intercept:

Interpret the fixed effect of linear time:

Interpret the fixed effect of quadratic time:

Interpret the effect of Age80:

Interpret the effect of linear*Age80:

Interpret the effect of quadratic*Age80:

Is the age by quadratic model (2b) better than the unconditional quadratic growth model (2a)? How do we know?

Contrasts

| Label | Num DF | Den DF | Chi-Square | F Value | Pr > ChiSq | Pr > F |
|---|---|---|---|---|---|---|
| DF=3 Wald Test for Effects of Age | 3 | 99 | 12.01 | 4.00 | 0.0073 | 0.0098 |

### Syntax and output from additional macros for effect size:

```
* Calculate Total R2 change relative to unconditional model;
%TotalR2(DV=nm3rt, PredFewer=PredQUnc, PredMore=PredQAge);
```

**Total R2 (% Reduction) for PredQUnc vs. PredQAge**

| Name | Pred Corr | TotalR2 | Total R2Diff |
|---|---|---|---|
| PredQUnc | 0.19167 | 0.03674 | . |
| **PredQAge** | **0.32688** | **0.10685** | **0.070114** |

```
* Calculate PseudoR2 relative to unconditional model;
%PseudoR2(NCov=7, CovFewer=CovQUnc, CovMore=CovQAge);
```

**PsuedoR2 (% Reduction) for CovQUnc vs. CovQAge**

| Name | CovParm | Subject | Estimate | StdErr | ZValue | ProbZ | Pseudo R2 |
|---|---|---|---|---|---|---|---|
| CovQUnc | UN(1,1) | ID | 276206 | 41442 | 6.66 | <.0001 | . |
| CovQUnc | UN(2,2) | ID | 25840 | 5864.41 | 4.41 | <.0001 | . |
| CovQUnc | UN(3,3) | ID | 634.47 | 172.37 | 3.68 | 0.0001 | . |
| CovQUnc | session | ID | 20298 | 1649.11 | 12.31 | <.0001 | . |
| CovQAge | UN(1,1) | ID | 247691 | 37599 | 6.59 | <.0001 | 0.10324 |
| CovQAge | UN(2,2) | ID | 25083 | 5787.37 | 4.33 | <.0001 | 0.02931 |
| CovQAge | UN(3,3) | ID | 629.58 | 172.51 | 3.65 | 0.0001 | 0.00770 |
| CovQAge | session | ID | 20298 | 1649.11 | 12.31 | <.0001 | 0.00000 |

Which variance component should have been reduced by each new fixed effect of age?

---

### Model 2c. Quadratic Model with Age, Reasoning on Intercept, Linear, Quadratic Time Slopes

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Session}_{ti} - 1\right) + \beta_{2i}\left(\text{Session}_{ti} - 1\right)^2 + e_{ti}$

Level 2:

Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01}\left(\text{Age}_i - 80\right) + \gamma_{02}\left(\text{Reason}_i - 22\right) + U_{0i}$

Linear: $\beta_{1i} = \gamma_{10} + \gamma_{11}\left(\text{Age}_i - 80\right) + \gamma_{12}\left(\text{Reason}_i - 22\right) + U_{1i}$

Quadratic: $\beta_{2i} = \gamma_{20} + \gamma_{21}\left(\text{Age}_i - 80\right) + \gamma_{22}\left(\text{Reason}_i - 22\right) + U_{2i}$

```
TITLE1 "2c: Keep Age, Add Reasoning on Intercept, Linear, and Quadratic Time Slopes";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=REML;
       CLASS ID session;
       MODEL nm3rt = time1 time1*time1 Age80 time1*Age80 time1*time1*Age80
                     Reas22 time1*Reas22 time1*time1*Reas22
                    / SOLUTION DDFM=Satterthwaite OUTPM=PredQReas; * Save fixed-predicted RT;
       RANDOM INTERCEPT time1 time1*time1 / GCORR TYPE=UN SUBJECT=ID;
       REPEATED session / TYPE=VC SUBJECT=ID;
```

```
       ODS OUTPUT CovParms=CovQReas; * Save covparms for comparison;
       CONTRAST "DF=3 Wald Test for Effects of Age"
                Age80 1, Age80*time1 1, Age80*time1*time1 1 / CHISQ;
       CONTRAST "DF=3 Wald Test for Effects of Reasoning"
                Reas22 1, Reas22*time1 1, Reas22*time1*time1 1 / CHISQ;
* Requesting additional effects for reasoning;
ESTIMATE "Reasoning Effect at Session 1"   Reas22 1 time1*Reas22 0 time1*time1*Reas22 0;
ESTIMATE "Reasoning Effect at Session 2"   Reas22 1 time1*Reas22 1 time1*time1*Reas22 1;
ESTIMATE "Reasoning Effect at Session 3"   Reas22 1 time1*Reas22 2 time1*time1*Reas22 4;
ESTIMATE "Reasoning Effect at Session 4"   Reas22 1 time1*Reas22 3 time1*time1*Reas22 9;
ESTIMATE "Reasoning Effect at Session 5"   Reas22 1 time1*Reas22 4 time1*time1*Reas22 16;
ESTIMATE "Reasoning Effect at Session 6"   Reas22 1 time1*Reas22 5 time1*time1*Reas22 25;
ESTIMATE "Reasoning*Linear Time Slope at Session 1" time1*Reas22 1 time1*time1*Reas22 0;
ESTIMATE "Reasoning*Linear Time Slope at Session 2" time1*Reas22 1 time1*time1*Reas22 2;
ESTIMATE "Reasoning*Linear Time Slope at Session 3" time1*Reas22 1 time1*time1*Reas22 4;
ESTIMATE "Reasoning*Linear Time Slope at Session 4" time1*Reas22 1 time1*time1*Reas22 6;
ESTIMATE "Reasoning*Linear Time Slope at Session 5" time1*Reas22 1 time1*time1*Reas22 8;
ESTIMATE "Reasoning*Linear Time Slope at Session 6" time1*Reas22 1 time1*time1*Reas22 10;
RUN; TITLE1;
```

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr Z |
|---|---|---|---|---|---|
| UN(1,1) | ID | 235541 | 36056 | 6.53 | <.0001 |
| UN(2,1) | ID | -32552 | 11138 | -2.92 | 0.0035 |
| UN(2,2) | ID | 25228 | 5835.93 | 4.32 | <.0001 |
| UN(3,1) | ID | 3918.44 | 1826.88 | 2.14 | 0.0320 |
| UN(3,2) | ID | -3812.99 | 978.05 | -3.90 | <.0001 |
| UN(3,3) | ID | 614.47 | 171.25 | 3.59 | 0.0002 |
| session | ID | 20298 | 1649.11 | 12.31 | <.0001 |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| 8261.0 | 7 | 8275.0 | 8275.2 | 8282.4 | 8293.3 | 8300.3 |

Solution for Fixed Effects

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Intercept | 1966.47 | 50.4203 | 98 | 39.00 | <.0001 |
| time1 | -119.74 | 20.0746 | 98 | -5.96 | <.0001 |
| time1*time1 | 13.3036 | 3.4167 | 98 | 3.89 | 0.0002 |
| Age80 | 22.2782 | 8.7324 | 98 | 2.55 | 0.0123 |
| time1*Age80 | -6.4921 | 3.4768 | 98 | -1.87 | 0.0649 |
| time1*time1*Age80 | 0.9601 | 0.5917 | 98 | 1.62 | 0.1079 |
| Reas22 | -27.1004 | 11.2829 | 98 | -2.40 | 0.0182 |
| time1*Reas22 | -3.5917 | 4.4922 | 98 | -0.80 | 0.4259 |
| time1*time1*Reas22 | 1.1575 | 0.7646 | 98 | 1.51 | 0.1333 |

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Reasoning Effect at Session 1 | -27.1004 | 11.2829 | 98 | -2.40 | 0.0182 |
| Reasoning Effect at Session 2 | -29.5346 | 10.1156 | 98 | -2.92 | 0.0043 |
| Reasoning Effect at Session 3 | -29.6537 | 9.8944 | 98 | -3.00 | 0.0035 |
| Reasoning Effect at Session 4 | -27.4578 | 9.7730 | 98 | -2.81 | 0.0060 |
| Reasoning Effect at Session 5 | -22.9468 | 9.5224 | 98 | -2.41 | 0.0178 |
| Reasoning Effect at Session 6 | -16.1207 | 9.6403 | 98 | -1.67 | 0.0977 |
| Reasoning*Linear Time Slope at Session 1 | -3.5917 | 4.4922 | 98 | -0.80 | 0.4259 |
| Reasoning*Linear Time Slope at Session 2 | -1.2767 | 3.0547 | 98 | -0.42 | 0.6769 |
| Reasoning*Linear Time Slope at Session 3 | 1.0384 | 1.7775 | 98 | 0.58 | 0.5604 |
| Reasoning*Linear Time Slope at Session 4 | 3.3535 | 1.2900 | 98 | 2.60 | 0.0108 |
| Reasoning*Linear Time Slope at Session 5 | 5.6686 | 2.2012 | 98 | 2.58 | 0.0115 |
| Reasoning*Linear Time Slope at Session 6 | 7.9836 | 3.5642 | 98 | 2.24 | 0.0274 |




| Label | Contrasts Num DF | Den DF | Chi-Square | F Value | Pr > ChiSq | Pr > F |
|---|---|---|---|---|---|---|
| DF=3 Wald Test for Effects of Age | 3 | 98 | 7.19 | 2.40 | 0.0660 | 0.0727 |
| DF=3 Wald Test for Effects of Reasoning | 3 | 98 | 12.88 | 4.29 | 0.0049 | 0.0068 |

## Syntax and output from additional macros for effect size:

```
* Calculate Total R2 change relative to age only model;
%TotalR2(DV=nm3rt, PredFewer=PredQAge, PredMore=PredQReas);
```

**Total R2 (% Reduction) for PredQAge vs. PredQReas**

| Name | Pred Corr | TotalR2 | Total R2Diff |
|---|---|---|---|
| PredQAge | 0.32688 | 0.10685 | . |
| PredQReas | 0.40108 | 0.16086 | 0.054011 |

```
* Calculate PseudoR2 relative to age only model;
%PseudoR2(NCov=7, CovFewer=CovQAge, CovMore=CovQReas);
```

**PsuedoR2 (% Reduction) for CovQAge vs. CovQReas**

| Name | CovParm | Subject | Estimate | StdErr | ZValue | ProbZ | PseudoR2 |
|---|---|---|---|---|---|---|---|
| CovQAge | UN(1,1) | ID | 247691 | 37599 | 6.59 | <.0001 | . |
| CovQAge | UN(2,2) | ID | 25083 | 5787.37 | 4.33 | <.0001 | . |
| CovQAge | UN(3,3) | ID | 629.58 | 172.51 | 3.65 | 0.0001 | . |
| CovQAge | session | ID | 20298 | 1649.11 | 12.31 | <.0001 | . |
| CovQReas | UN(1,1) | ID | 235541 | 36056 | 6.53 | <.0001 | 0.049052 |
| CovQReas | UN(2,2) | ID | 25228 | 5835.93 | 4.32 | <.0001 | -0.005808 |
| CovQReas | UN(3,3) | ID | 614.47 | 171.25 | 3.59 | 0.0002 | 0.024008 |
| CovQReas | session | ID | 20298 | 1649.11 | 12.31 | <.0001 | -0.000000 |

From these results *it appears* we could remove both the interaction of reasoning with both the linear and quadratic time slopes, but keep in mind how correlated those terms are… let's see what happens if we just remove just the reasoning*quadratic time interaction for now.

## Model 2d. Quadratic Model without Reasoning by Quadratic Time Slope

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Session}_{ti} - 1\right) + \beta_{2i}\left(\text{Session}_{ti} - 1\right)^2 + e_{ti}$

Level 2:

Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01}\left(\text{Age}_i - 80\right) + \gamma_{02}\left(\text{Reason}_i - 22\right) + U_{0i}$

Linear: $\beta_{1i} = \gamma_{10} + \gamma_{11}\left(\text{Age}_i - 80\right) + \gamma_{12}\left(\text{Reason}_i - 22\right) + U_{1i}$

Quadratic: $\beta_{2i} = \gamma_{20} + \gamma_{21}\left(\text{Age}_i - 80\right) + U_{2i}$

```
TITLE1 "2d: Remove Reasoning Effect on Quadratic Time Slope";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=REML;
       CLASS ID session;
       MODEL nm3rt = time1 time1*time1 Age80 time1*Age80 time1*time1*Age80 Reas22 time1*Reas22
                     / SOLUTION DDFM=Satterthwaite OUTPM=PredLReas; * Save fixed-predicted RT;
       RANDOM INTERCEPT time1 time1*time1 / GCORR TYPE=UN SUBJECT=ID;
       REPEATED session / TYPE=VC SUBJECT=ID;
       ODS OUTPUT CovParms=CovLReas; * Save covparms for comparison;
       CONTRAST "DF=3 Wald Test for Effects of Age"
                Age80 1, Age80*time1 1, Age80*time1*time1 1 / CHISQ;
       CONTRAST "DF=2 Wald Test for Effects of Reasoning" Reas22 1, Reas22*time1 1 / CHISQ;
* Requesting additional effects for reasoning;
ESTIMATE "Reasoning Effect at Session 1" Reas22 1 time1*Reas22 0;
ESTIMATE "Reasoning Effect at Session 2" Reas22 1 time1*Reas22 1;
ESTIMATE "Reasoning Effect at Session 3" Reas22 1 time1*Reas22 2;
ESTIMATE "Reasoning Effect at Session 4" Reas22 1 time1*Reas22 3;
ESTIMATE "Reasoning Effect at Session 5" Reas22 1 time1*Reas22 4;
ESTIMATE "Reasoning Effect at Session 6" Reas22 1 time1*Reas22 5; RUN; TITLE1;
```

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr Z |
|---|---|---|---|---|---|
| UN(1,1) | ID | 235909 | 36153 | 6.53 | <.0001 |
| UN(2,1) | ID | -32972 | 11262 | -2.93 | 0.0034 |
| UN(2,2) | ID | 25707 | 5883.65 | 4.37 | <.0001 |
| UN(3,1) | ID | 3993.04 | 1848.58 | 2.16 | 0.0308 |
| UN(3,2) | ID | -3897.93 | 985.52 | -3.96 | <.0001 |
| UN(3,3) | ID | 629.52 | 172.50 | 3.65 | 0.0001 |
| session | ID | 20298 | 1649.11 | 12.31 | <.0001 |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| 8264.6 | 7 | 8278.6 | 8278.8 | 8286.0 | 8296.9 | 8303.9 |

Solution for Fixed Effects

| Effect | Estimate | Standard Error | DF | t Value | Pr > \|t\| | |
|---|---|---|---|---|---|---|
| Intercept | 1969.80 | 50.4084 | 98.1 | 39.08 | <.0001 | |
| time1 | -123.54 | 20.0358 | 98.9 | -6.17 | <.0001 | |
| time1*time1 | 13.9774 | 3.4095 | 99 | 4.10 | <.0001 | |
| Age80 | 20.8470 | 8.6868 | 99.7 | 2.40 | 0.0183 | |
| time1*Age80 | -4.8610 | 3.3252 | 100 | -1.46 | 0.1469 | |
| time1*time1*Age80 | 0.6709 | 0.5637 | 99 | 1.19 | 0.2368 | |
| Reas22 | -32.8281 | 10.6297 | 98 | -3.09 | 0.0026 | |
| time1*Reas22 | 2.9362 | 1.2602 | 98 | 2.33 | 0.0219 | → Different result! |

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| Reasoning Effect at Session 1 | -32.8281 | 10.6297 | 98 | -3.09 | 0.0026 |

| | | | | | |
|---|---|---|---|---|---|
| Reasoning Effect at Session 2 | -29.8919 | 10.1128 | 98 | -2.96 | 0.0039 |
| Reasoning Effect at Session 3 | -26.9557 | 9.7326 | 98 | -2.77 | 0.0067 |
| Reasoning Effect at Session 4 | -24.0195 | 9.5055 | 98 | -2.53 | 0.0131 |
| Reasoning Effect at Session 5 | -21.0833 | 9.4425 | 98 | -2.23 | 0.0278 |
| Reasoning Effect at Session 6 | -18.1471 | 9.5469 | 98 | -1.90 | 0.0603 |

| | Contrasts | | | | | |
|---|---|---|---|---|---|---|
| | Num | Den | | | | |
| Label | DF | DF | Chi-Square | F Value | Pr > ChiSq | Pr > F |
| DF=3 Wald Test for Effects of Age | 3 | 100 | 5.98 | 1.99 | 0.1128 | 0.1200 |
| DF=2 Wald Test for Effects of Reasoning | 2 | 98 | 10.59 | 5.29 | 0.0050 | 0.0066 |




## Syntax and output from additional macros for effect size:

```
* Calculate Total R2 change relative to age only model;
%TotalR2(DV=nm3rt, PredFewer=PredQAge, PredMore=PredLReas);
```

**Total R2 (% Reduction) for PredQAge vs. PredLReas**

| Name | Pred Corr | TotalR2 | Total R2Diff |
|---|---|---|---|
| PredQAge | 0.32688 | 0.10685 | . |
| PredLReas | 0.40008 | 0.16006 | 0.053213 |

```
* Calculate PseudoR2 relative to age model;
%PseudoR2(NCov=7, CovFewer=CovQAge, CovMore=CovQReas);
```

**PsuedoR2 (% Reduction) for CovQAge vs. CovLReas**

| Name | CovParm | Subject | Estimate | StdErr | ZValue | ProbZ | PseudoR2 |
|---|---|---|---|---|---|---|---|
| CovQAge | UN(1,1) | ID | 247691 | 37599 | 6.59 | <.0001 | . |
| CovQAge | UN(2,2) | ID | 25083 | 5787.37 | 4.33 | <.0001 | . |
| CovQAge | UN(3,3) | ID | 629.58 | 172.51 | 3.65 | 0.0001 | . |
| CovQAge | session | ID | 20298 | 1649.11 | 12.31 | <.0001 | . |
| CovLReas | UN(1,1) | ID | 235909 | 36153 | 6.53 | <.0001 | 0.047565 |
| CovLReas | UN(2,2) | ID | 25707 | 5883.65 | 4.37 | <.0001 | -0.024908 |
| CovLReas | UN(3,3) | ID | 629.52 | 172.50 | 3.65 | 0.0001 | 0.000095 |
| CovLReas | session | ID | 20298 | 1649.11 | 12.31 | <.0001 | -0.000000 |

## Model 2e. Quadratic Model adding Effects of Education Group on Intercept, Linear, Quadratic Time

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Session}_{ti} - 1\right) + \beta_{2i}\left(\text{Session}_{ti} - 1\right)^2 + e_{ti}$

Level 2:

Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01}\left(\text{Age}_i - 80\right) + \gamma_{02}\left(\text{Reason}_i - 22\right) + \gamma_{03}\left(\text{Highvs.LowEd}_i\right) + \gamma_{04}\left(\text{Highvs.MedEd}_i\right) + U_{0i}$

Linear: $\beta_{1i} = \gamma_{10} + \gamma_{11}\left(\text{Age}_i - 80\right) + \gamma_{12}\left(\text{Reason}_i - 22\right) + \gamma_{13}\left(\text{Highvs.LowEd}_i\right) + \gamma_{14}\left(\text{Highvs.MedEd}_i\right) + U_{1i}$

Quadratic: $\beta_{2i} = \gamma_{20} + \gamma_{21}\left(\text{Age}_i - 80\right) + \gamma_{23}\left(\text{Highvs.LowEd}_i\right) + \gamma_{24}\left(\text{Highvs.MedEd}_i\right) + U_{2i}$

```
TITLE1 "2e: Keep Age & Reas, Add Effect of Education Group on Intercept, Linear, and Quadratic";
PROC MIXED DATA=&datafile. NOCLPRINT COVTEST IC NAMELEN=100 METHOD=REML;
       CLASS ID EducGrp session;
       MODEL nm3rt = time1 time1*time1 Age80 time1*Age80 time1*time1*Age80
                     Reas22 time1*Reas22 EducGrp time1*EducGrp time1*time1*EducGrp
                     / SOLUTION DDFM=Satterthwaite  OUTPM=PredQEduc; * Save fixed-predicted RT;
       RANDOM INTERCEPT time1 time1*time1 / GCORR TYPE=UN SUBJECT=ID;
       REPEATED session / TYPE=VC SUBJECT=ID;
       ODS OUTPUT CovParms=CovQEduc; * Save covparms for comparison;
       CONTRAST "DF=3 Wald Test for Effects of Age"
                 Age80 1, Age80*time1 1, Age80*time1*time1 1 / CHISQ;
       CONTRAST "DF=2 Wald Test for Effects of Reasoning" Reas22 1, Reas22*time1 1 / CHISQ;
       CONTRAST "DF=6 Wald Test for Effects of Education" EducGrp -1 1 0, EducGrp -1 0 1,
                  EducGrp*time1 -1 1 0, EducGrp*time1 -1 0 1,
                 EducGrp*time1*time1 -1 1 0, EducGrp*time1*time1 -1 0 1 / CHISQ;
       * LSMEANS gives follow-up tests and means per group for education main effect only;
       LSMEANS EducGrp / AT (time1 Age80 Reas22) = (0 0 0) DIFF=ALL;
       LSMEANS EducGrp / AT (time1 Age80 Reas22) = (5 0 0) DIFF=ALL;
       * ESTIMATE statements can also give specific effects as before;
       ESTIMATE "L vs. H Educ for Intercept "         EducGrp -1  0  1 ;
       ESTIMATE "M vs. H Educ for Intercept"          EducGrp  0 -1  1 ;
       ESTIMATE "L vs. M Educ for Intercept"          EducGrp -1  1  0 ;
       ESTIMATE "L vs. H Educ for Linear Time"        time1*EducGrp -1  0  1 ;
       ESTIMATE "M vs. H Educ for Linear Time"        time1*EducGrp  0 -1  1 ;
       ESTIMATE "L vs. M Educ for Linear Time"        time1*EducGrp -1  1  0 ;
       ESTIMATE "L vs. H Educ for Quadratic Time"   time1*time1*EducGrp -1  0  1 ;
       ESTIMATE "M vs. H Educ for Quadratic Time"   time1*time1*EducGrp  0 -1  1 ;
       ESTIMATE "L vs. M Educ for Quadratic Time"   time1*time1*EducGrp -1  1  0 ;
RUN; TITLE1;
```

Covariance Parameter Estimates

| Cov Parm | Subject | Estimate | Standard Error | Z Value | Pr Z |
|---|---|---|---|---|---|
| UN(1,1) | ID | 241027 | 37339 | 6.46 | <.0001 |
| UN(2,1) | ID | -35271 | 11645 | -3.03 | 0.0025 |
| UN(2,2) | ID | 25772 | 5956.96 | 4.33 | <.0001 |
| UN(3,1) | ID | 4371.57 | 1907.59 | 2.29 | 0.0219 |
| UN(3,2) | ID | -3896.53 | 995.30 | -3.91 | <.0001 |
| UN(3,3) | ID | 628.15 | 173.93 | 3.61 | 0.0002 |
| session | ID | 20298 | 1649.11 | 12.31 | <.0001 |

Information Criteria

| Neg2LogLike | Parms | AIC | AICC | HQIC | BIC | CAIC |
|---|---|---|---|---|---|---|
| 8211.4 | 7 | 8225.4 | 8225.6 | 8232.8 | 8243.7 | 8250.7 |

Solution for Fixed Effects

| Effect | Education Group (1=HS,2=BA,3=GRAD) | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|---|
| Intercept | | 1961.89 | 104.34 | 95.7 | 18.80 | <.0001 |
| time1 | | -106.50 | 41.1184 | 96.7 | -2.59 | 0.0111 |
| time1*time1 | | 12.4797 | 6.9879 | 97 | 1.79 | 0.0772 |

| | | | | | | |
|---|---|---|---|---|---|---|
| Age80 | | 20.2894 | 8.7750 | 97.5 | 2.31 | 0.0229 |
| time1*Age80 | | -4.5759 | 3.3351 | 98 | -1.37 | 0.1732 |
| time1*time1*Age80 | | 0.6177 | 0.5646 | 97 | 1.09 | 0.2767 |
| Reas22 | | -36.6221 | 11.0407 | 96 | -3.32 | 0.0013 |
| time1*Reas22 | | 2.9786 | 1.3130 | 96.1 | 2.27 | 0.0255 |
| EducGrp | 1 | -51.3792 | 154.85 | 96.3 | -0.33 | 0.7408 |
| EducGrp | 2 | 37.6426 | 123.90 | 95.4 | 0.30 | 0.7619 |
| EducGrp | 3 | 0 | . | . | . | . |
| time1*EducGrp | 1 | -70.2451 | 60.3032 | 97.1 | -1.16 | 0.2469 |
| time1*EducGrp | 2 | -4.3577 | 49.1299 | 96.5 | -0.09 | 0.9295 |
| time1*EducGrp | 3 | 0 | . | . | . | . |
| time1*time1*EducGrp | 1 | 11.0653 | 10.2358 | 97 | 1.08 | 0.2824 |
| time1*time1*EducGrp | 2 | -1.4641 | 8.3545 | 97 | -0.18 | 0.8612 |
| time1*time1*EducGrp | 3 | 0 | . | . | . | . |

Type 3 Tests of Fixed Effects

| Effect | Num DF | Den DF | F Value | Pr > F |
|---|---|---|---|---|
| time1 | 1 | 96.5 | 35.77 | <.0001 |
| time1*time1 | 1 | 97 | 17.62 | <.0001 |
| Age80 | 1 | 97.5 | 5.35 | 0.0229 |
| time1*Age80 | 1 | 98 | 1.88 | 0.1732 |
| time1*time1*Age80 | 1 | 97 | 1.20 | 0.2767 |
| Reas22 | 1 | 96 | 11.00 | 0.0013 |
| time1*Reas22 | 1 | 96.1 | 5.15 | 0.0255 |
| **EducGrp** | **2** | **96.1** | **0.23** | **0.7965** |
| **time1*EducGrp** | **2** | **97** | **0.92** | **0.4012** |
| **time1*time1*EducGrp** | **2** | **97** | **1.05** | **0.3545** |

I normally skip this box if the CLASS statement is not used for predictors, but here the last three entries give us the omnibus (df=2) tests for whether there are any education group differences on the intercept, linear, or quadratic time slopes, not just pairwise comparisons.

Estimates

| Label | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|
| L vs. H Educ for Intercept | 51.3792 | 154.85 | 96.3 | 0.33 | 0.7408 |
| M vs. H Educ for Intercept | -37.6426 | 123.90 | 95.4 | -0.30 | 0.7619 |
| L vs. M Educ for Intercept | 89.0218 | 134.02 | 96.8 | 0.66 | 0.5081 |
| L vs. H Educ for Linear Time | 70.2451 | 60.3032 | 97.1 | 1.16 | 0.2469 |
| M vs. H Educ for Linear Time | 4.3577 | 49.1299 | 96.5 | 0.09 | 0.9295 |
| L vs. M Educ for Linear Time | 65.8874 | 51.7661 | 97.4 | 1.27 | 0.2061 |
| L vs. H Educ for Quadratic Time | -11.0653 | 10.2358 | 97 | -1.08 | 0.2824 |
| M vs. H Educ for Quadratic Time | 1.4641 | 8.3545 | 97 | 0.18 | 0.8612 |
| L vs. M Educ for Quadratic Time | -12.5294 | 8.7793 | 97 | -1.43 | 0.1567 |

Contrasts

| Label | Num DF | Den DF | Chi-Square | F Value | Pr > ChiSq | Pr > F |
|---|---|---|---|---|---|---|
| DF=3 Wald Test for Effects of Age | 3 | 98.1 | 5.49 | 1.83 | 0.1395 | 0.1469 |
| DF=2 Wald Test for Effects of Reasoning | 2 | 96 | 11.70 | 5.85 | 0.0029 | 0.0040 |
| DF=6 Wald Test for Effects of Education | 6 | 96.4 | 4.59 | 0.76 | 0.5976 | 0.5994 |

Least Squares Means

| Effect | Educ Grp | time1 | Age80 | Reas22 | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|---|---|---|---|
| EducGrp | 1 | 0.00 | 0.00 | 0.00 | 1910.51 | 112.41 | 96.1 | 17.00 | <.0001 |
| EducGrp | 2 | 0.00 | 0.00 | 0.00 | 1999.53 | 69.2506 | 96.3 | 28.87 | <.0001 |
| EducGrp | 3 | 0.00 | 0.00 | 0.00 | 1961.89 | 104.34 | 95.7 | 18.80 | <.0001 |
| EducGrp | 1 | 5.00 | 0.00 | 0.00 | 1615.41 | 95.7317 | 96 | 16.87 | <.0001 |
| EducGrp | 2 | 5.00 | 0.00 | 0.00 | 1720.63 | 59.0105 | 96.1 | 29.16 | <.0001 |
| EducGrp | 3 | 5.00 | 0.00 | 0.00 | 1741.38 | 88.7887 | 95.9 | 19.61 | <.0001 |

In LSMEANS, you must specify a value at which to hold each continuous predictor.

Differences of Least Squares Means

| Effect | Educ Grp | Educ Grp | time1 | Age80 | Reas22 | Estimate | Standard Error | DF | t Value | Pr > \|t\| |
|---|---|---|---|---|---|---|---|---|---|---|
| EducGrp | 1 | 2 | 0.00 | 0.00 | 0.00 | -89.0218 | 134.02 | 96.8 | -0.66 | 0.5081 |
| EducGrp | 1 | 3 | 0.00 | 0.00 | 0.00 | -51.3792 | 154.85 | 96.3 | -0.33 | 0.7408 |
| EducGrp | 2 | 3 | 0.00 | 0.00 | 0.00 | 37.6426 | 123.90 | 95.4 | 0.30 | 0.7619 |
| EducGrp | 1 | 2 | 5.00 | 0.00 | 0.00 | -105.22 | 114.33 | 96.2 | -0.92 | 0.3597 |
| EducGrp | 1 | 3 | 5.00 | 0.00 | 0.00 | -125.97 | 131.97 | 96.1 | -0.95 | 0.3422 |
| EducGrp | 2 | 3 | 5.00 | 0.00 | 0.00 | -20.7486 | 105.35 | 95.9 | -0.20 | 0.8443 |

## Syntax and output from additional macros for effect size:

```
* Calculate Total R2 change relative to model with reasoning*linear only;
%TotalR2(DV=nm3rt, PredFewer=PredLReas, PredMore=PredQEduc);
```

**Total R2 (% Reduction) for PredLReas vs. PredQEduc**

| Name | Pred Corr | TotalR2 | Total R2Diff |
|---|---|---|---|
| PredLReas | 0.40008 | 0.16006 | . |
| PredQEduc | 0.41510 | 0.17231 | 0.012242 |

```
* Calculate PseudoR2 relative to model with reasoning*linear only;
%PseudoR2(NCov=7, CovFewer=CovLReas, CovMore=CovQEduc);
```

**PsuedoR2 (% Reduction) for CovLReas vs. CovQEduc**

| Name | CovParm | Subject | Estimate | StdErr | ZValue | ProbZ | PseudoR2 |
|---|---|---|---|---|---|---|---|
| CovLReas | UN(1,1) | ID | 235909 | 36153 | 6.53 | <.0001 | . |
| CovLReas | UN(2,2) | ID | 25707 | 5883.65 | 4.37 | <.0001 | . |
| CovLReas | UN(3,3) | ID | 629.52 | 172.50 | 3.65 | 0.0001 | . |
| CovLReas | session | ID | 20298 | 1649.11 | 12.31 | <.0001 | . |
| CovQEduc | UN(1,1) | ID | 241027 | 37339 | 6.46 | <.0001 | -0.021693 |
| CovQEduc | UN(2,2) | ID | 25772 | 5956.96 | 4.33 | <.0001 | -0.002519 |
| CovQEduc | UN(3,3) | ID | 628.15 | 173.93 | 3.61 | 0.0002 | 0.002185 |
| CovQEduc | session | ID | 20298 | 1649.11 | 12.31 | <.0001 | -0.000000 |

Based on the lack of significance of the effect of education, I'd say we're done with this model (I had previously tried age*reasoning, and none of those higher-order effects were significant).
The age*quadratic interaction could probably be removed, but I choose to leave it in as a control.

# Simple Processing Speed: Example Conditional Models of Change Results

The extent to which individual differences in response time (RT) over six sessions for a simple processing speed test (number match three) could be predicted from baseline age, abstract reasoning, and education level was examined in a series of multilevel models (i.e., general linear mixed models) in which the six practice sessions were nested within each participant. Residual maximum likelihood (REML) was used in estimating and reporting all model parameters; denominator degrees of freedom were estimated using the Satterthwaite method. The significance of new fixed effects were evaluated with univariate and multivariate Wald tests. Session (i.e., the index of time) was centered at the first occasion, age was centered at 80 years, abstract reasoning was centered at 22 (near the mean of the scale), and graduate-level education was the reference group for education level (with separate contrasts for high school or less and for bachelor's level education).

**Piecewise Time Models**

The best-fitting unconditional growth model specified linear decline from sessions 1–2 and a second, more shallow rate of linear decline from sessions 2–6, along with significant individual differences in the intercept and in each linear slope. Accordingly, effect size was evaluated via pseduo-$R^2$ values for the proportion reduction in each random effect variance, as well as with total-$R^2$, the squared correlation between the actual outcome values and the outcomes predicted by the model fixed effects. In the unconditional growth model, the fixed effects for linear and quadratic change across sessions accounted for approximately 4% of the total variation in RT.

Next, age was added as a predictor of the intercept and each linear slope. Although the three effects of age together resulted in a significant omnibus effect, $F(3, 99) = 4.08$, $p < .01$, only the fixed effect of age on the intercept was significant, indicating that for every additional year of age above 80, RT at the first session was predicted to be significantly higher (slower) by 29.78 ($p < .001$). In terms of pseudo-$R^2$, age accounted for 10.56% of the level-2 random intercept variance, 1.90% of the level-2 random variance in linear change from sessions 1–2, and 0.91% of the level-2 random variance in linear change from sessions 2–6. As expected given that baseline age is a time-invariant predictor, the level-1 residual variance was not reduced. The cumulative total-$R^2$ from session and age was $R^2 = .11$, approximately a 7% increase due to age. Although the interactions of age with the linear piecewise slopes were not significant, they were retained in the model to fully control for age effects before examining the other predictors.

Abstract reasoning was then added as a predictor of the intercept and each linear slope. The three effects of abstract reasoning together resulted in a significant omnibus effect, $F(3, 98) = 3.50$, $p = .02$. The significant fixed effects of abstract reasoning on the intercept and first slope indicated that for every additional unit of reasoning above 22, RT at the first session was predicted to be significantly lower (faster) by 27.10 ($p < .001$) and to decrease by an additional 4.71 ms by session 2.The nonsignificant effect of reasoning on the second slope was retained to facilitate interpretation of the separate effects of reasoning on each aspect of change. Reasoning accounted for 4.76% of the level-2 random intercept variance, none of the level-2 random first slope variance, and 0.70% of the level-2 second slope variance. The cumulative total-$R^2$ from session, age, and reasoning was 16%, approximately a 5% increase due to reasoning.

Education level (high school or less, bachelor's level, or graduate level) was then added as a predictor of the intercept and each linear slope. These six effects of education did not significantly improve model fit, $F(6, 96) = 0.73$, $p = .63$. No omnibus main effects of education level on the intercept, linear, or quadratic slopes were significant, and no pairwise comparisons were significant as well. Education accounted for no measurable variance in the level-2 random intercept or either level-2 random linear slope. The cumulative $R^2$ from session, age, reasoning, and education was total-$R^2 = .17$, approximately a 1% increase due to education.

Finally, we examined the interactive effects of age and reasoning in predicting the intercept and each linear slope, although none was significant.

(From here one might remove nonsignificant model effects and/or add other effects as needed to fully answer all research questions…)

**Quadratic Time Models**

The best-fitting unconditional growth model specified quadratic decline across the six sessions (i.e., a decelerating negative function) with significant individual differences in the intercept, linear, and quadratic time effects. Accordingly, effect size was evaluated via pseduo-$R^2$ values for the proportion reduction in each random effect variance, as well as with total-$R^2$, the squared correlation between the actual outcome values and the outcomes predicted by the model fixed effects. In the unconditional growth model, the fixed effects for linear and quadratic change across sessions accounted for approximately 4% of the total variation in RT.

Next, age was added as a predictor of the intercept, linear slope, and quadratic slope. Although the three effects of age together resulted in a significant omnibus effect, $F(3, 99) = 4.00$, $p < .01$, only the fixed effect of age on the intercept was significant, indicating that for every additional year of age above 80, RT at the first session was predicted to be significantly higher (slower) by 29.05 ($p < .001$). In terms of pseudo-$R^2$, age accounted for 10.32% of the level-2 random intercept variance, 2.93% of the level-2 random linear slope variance, and 0.77% of the level-2 random quadratic slope variance. As expected given that baseline age is a time-invariant predictor, the level-1 residual variance was not reduced. The cumulative total-$R^2$ from session and age was 11%, approximately a 7% increase due to age. Although the interactions of age with the linear and quadratic slopes were not significant, they were retained in the model to fully control for age effects before examining the effects of other predictors.

Abstract reasoning was then added as a predictor of the intercept, linear slope, and quadratic slope. As with the effects of age, although the three effects of abstract reasoning together resulted in a significant omnibus effect, $F(3, 98) = 4.29$, $p < .01$, only the fixed effect of abstract reasoning on the intercept was significant, indicating that for every additional unit of reasoning above 22, RT at the first session was predicted to be significantly lower (faster) by 27.10 ($p < .001$).The nonsignificant effect of reasoning on the quadratic slope was then removed, revealing a significant effect of reasoning on both the intercept and linear slope, $F(2, 98) = 5.29$, $p < .01$, such that for every unit higher reasoning above 22, RT at the first session was expected to be lower by 32.83 and the linear rate of improvement in RT (as evaluated at the first session given the quadratic slope) was expected to be less negative by 2.94 (i.e., faster initial RT with less improvement in persons with greater reasoning). Reasoning accounted for 4.76% of the level-2 random intercept variance but had no measurable reduction of the level-2 random linear and quadratic slope variances. The cumulative total-$R^2$ from session, age, and reasoning was 16%, approximately a 5% increase due to reasoning.

Education level (high school or less, bachelor's level, or graduate level) was then added as a predictor of the intercept, linear slope, and quadratic slope. These six effects of education did not significantly improve model fit, $F(6, 96) = 0.76$, $p = .60$. No omnibus main effects of education level on the intercept, linear, or quadratic slopes were significant, and no pairwise comparisons were significant as well. Education accounted for no measurable random intercept or random linear slope variance, and 2.19% of the random quadratic slope variance. The cumulative total-$R^2$ from session, age, reasoning, and education was 17%, approximately a 1% increase due to education.

Finally, we examined the interactive effects of age and reasoning in predicting the intercept and each linear slope, although none was significant.

(From here one might remove nonsignificant model effects and/or add other effects as needed to fully answer all research questions…)